# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgname=python-aaf2
pkgver=1.6.0
pkgrel=2
pkgdesc='Read and write Advanced Authoring Format (AAF) files'
url='http://pyaaf.readthedocs.io/'
license=(MIT)
arch=(any)
depends=(python)
makedepends=(python-build python-installer python-setuptools python-wheel)
source=(https://pypi.python.org/packages/source/p/pyaaf2/pyaaf2-$pkgver.tar.gz)
sha256sums=('83815554dbe56aabae4cf48d02e58cd7c102733d649672832ec8a2542d3f24ad')

build() {
  cd pyaaf2-$pkgver
  python -m build --wheel --no-isolation
}

package() {
  cd pyaaf2-$pkgver
  python -m installer --destdir="$pkgdir" dist/*.whl
  install -Dm644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname
}
